

Excel Yourself 2013

By Neale Blackwood

www.a4accounting.com.au

a4@iinet.net.au



Contents

Introduction	3
February 2013 - Summing the Top Five Values in a Range.....	4
May & December 2013 - AGGREGATE Function and a MAXIF Solution	6
July 2013 - Pareto Chart.....	9
September 2013 - NETWORKDAYS Function	11
October 2013 - Deleting Zero Rows Solutions	13
Appendix - Shortcuts.....	15

Introduction

This is my first webinar where I'll review some of my Excel Yourself articles that were first published in the INTHEBLACK magazine during the year. I won't be covering every single article as time is limited. I have chosen to cover those articles that will benefit most from the webinar format.

Six out of the eleven articles will be covered.

Depending on the response and feedback from the webinar I may do the other five articles in another webinar and also go back and review previous years' articles as well.

The workings sheets for all eleven articles have been included in the companion file for this session. This includes the code for the macro articles as well. The macro code can be viewed by pressing Alt + F11 and clicking on the module (left hand side) for the month involved.

The companion Excel sheets are arranged in date order starting with the February article on the left and working to the right to end with the December article.

In the Excel file the green sheet tabs are the ones to be covered during the session.

I have included some extra content in some of the articles.

The Appendix lists all of the shortcuts that were included at the end of the articles.

The articles were all first published in the CPA Australia INTHEBLACK magazine during 2013.

The content of this manual has been re-written and adapted from the original articles.

February 2013 - Summing the Top Five Values in a Range

This technique is more flexible than just the highest five. It can be extended to sum as many as required. Also you can easily modify it to sum a number of the lowest values as well.

Two functions are used together to achieve the result. Both of these functions can work with arrays. Arrays are an advanced concept in Excel. An array itself is just a list of values, but it is a powerful concept that can be used to do complicated calculations. We'll touch on them again later in the session when we look at the AGGREGATE function.

The two functions are the LARGE and SUMPRODUCT functions. The SUMPRODUCT is my favourite Excel function as it is Excel's most versatile and flexible function. The SUMPRODUCT has the ability to perform array calculations without having to use the array entry keys.

The LARGE function is like a flexible MAX functions. MAX only extracts the largest value but the LARGE function can extract the second, third, fourth or whatever position you want.

Syntax

LARGE(Range,Number)

Range - any range, row or column

Number - represents the position to extract, 1 represent the highest value, 3 is the third highest, etc.

Examples

The image below shows the values to be reviewed in the range A2:A15.

	A	B	C	D	E	F	G
1	Values		Third largest value	Formula in cell C2		1	99
2	28		96	=LARGE(A2:A15,3)		2	98
3	69					3	96
4	19					4	95
5	66		Sum top 5 values	Formula in cell C6		5	90
6	90		478	=SUMPRODUCT(LARGE(A2:A15,{1,2,3,4,5}))			478
7	99						
8	66						
9	95		Sum top 5 values	Alternative formula in cell C10			
10	37		478	=SUMPRODUCT(LARGE(A2:A15,ROW(1:5)))			
11	98						
12	14						
13	60		Sum top 10 values	Alternative formula in cell C14			
14	26		776	=SUMPRODUCT(LARGE(A2:A15,ROW(1:10)))			
15	96						

An example of the LARGE function is in G2.

=LARGE(\$A\$2:\$A\$15,F1)

The formula has been copied down. I've used the range F1:F5 to specify the Number to extract for the LARGE functions. You can see that the top 5 numbers add up to 478 in cell G6.

The SUMPRODUCT function means to SUM the PRODUCT of the values within its brackets. Product is a mathematical term for multiply. We use a hack for the function which isn't shown in the Excel's help system to perform array-type calculations.

Cell C6 has the formula that calculates the total of the top 5.

```
=SUMPRODUCT(LARGE(A2:A15,{1,2,3,4,5}))
```

The interesting part of this formula is the

```
{1,2,3,4,5}
```

This is called array syntax and it forces Excel to perform an array calculation. Most functions can't work with array syntax - they will return an error.

By using the array syntax in this case you force Excel to perform the following 5 separate LARGE calculations (which were calculated separately in cells G1:G5)

```
LARGE(A2:A15,1)
```

```
LARGE(A2:A15,2)
```

```
LARGE(A2:A15,3)
```

```
LARGE(A2:A15,4)
```

```
LARGE(A2:A15,5)
```

Those 5 separate results are then summed by the SUMPRODUCT to arrive at the result.

Typing in the array syntax doesn't scale very well. To do the top 10 double the number of entries.

If you need to SUM more numbers there is another technique that generates an array for you.

The technique uses the ROW function. The ROW function simply returns the row number for the reference within its brackets. If you leave out the reference it returns the row number of the row it is used in.

When you refer to a range within the ROW function it creates an array of the row numbers involved. To create our first five numbers you would use ROW(1:5). This has been used in the formula in cell C10.

```
=SUMPRODUCT(LARGE(A2:A15,ROW(1:5)))
```

This technique scales very well. The formula to SUM the top 10 values is only one digit longer and used in cell C14.

```
=SUMPRODUCT(LARGE(A2:A15,ROW(1:10)))
```

To SUM the 10 lowest values you would use

```
=SUMPRODUCT(SMALL(A2:A15,ROW(1:10)))
```

May & December 2013 - AGGREGATE Function and a MAXIF Solution

May 2013

The AGGREGATE function only works in Excel 2010 and later versions. It is the only function that allows you to ignore errors when performing calculations on ranges. The AGGREGATE function works in a similar way to the SUBTOTAL function in that it can ignore other AGGREGATE and SUBTOTAL functions within its range. Also like the SUBTOTAL it also has the ability to ignore hidden rows within its range.

The AGGREGATE function only works on columns of data, not on rows.

The AGGREGATE function is more powerful than the SUBTOTAL function because it can perform more functions. The SUBTOTAL function has the ability to perform most basic functions. The most commonly used is the SUM function, but it can perform COUNT and AVERAGE functions as well. The extra functions that the AGGREGATE function can perform are more advanced and one was demonstrated in the December 2013 article.

The May 2013 article handled the AGGREGATE function's ability to ignore errors in a range.

In the image on the right you can see all of Excel's error messages displayed in column A. Normally this would stop you being able to add up the values in that column because the SUM function would display an error message.

Whilst I don't encourage you to leave errors in your files if you do need to do a calculation on a range that contains errors then the AGGREGATE function can be used.

	A	B	C
1	Values		SUM
2	#DIV/0!		210
3	10		
4	#N/A		
5	20		
6	0		
7	30		
8	#NAME?		
9	40		
10	#NULL!		
11	50		
12	#NUM!		
13	60		
14	#REF!		

Syntax

AGGREGATE(Function,Options,Range,Number)

Function - a number that determines the function to use in the calculation.


Options - a number that specifies the treatment of the cells within the range. You can specify to ignore other SUBTOTAL and AGGREGATE functions within the range. You can specify to ignore hidden rows and you can specify to ignore error cells.

Range - the range to perform the calculation on.

Number (optional) - this is used for those functions that require an extra input.

As you build the AGGREGATE formula Excel will show you the various values for the entries.

See the following screenshots.



The final formula in cell C2 is

`=AGGREGATE(9,6,A2:A14)`

The 9 defines the SUM function and the 6 specifies ignore errors.

December 2013

The December 2013 article uses the LARGE function which we saw earlier used with the SUMPRODUCT function. As mentioned earlier the LARGE function has the ability to work with arrays. The AGGREGATE's ability to ignore errors allows you to create a formula that can create a conditional MAX function.

As at Excel 2013 there is no MAXIF function. A conditional MAX function would allow you to specify a condition and find the largest value that matches that condition. The AGGREGATE function allows you to create a formula that can work like a MAXIF would.

AGGREGATE's ability to ignore errors is combined with Excel's TRUE and FALSE syntax to create errors on those values that do not match the criteria.

In Excel TRUE equals one and FALSE equals zero. Comparing cells to each other returns TRUE if they are equal and FALSE if they are not. Hence dividing by FALSE will cause a divide by zero error. Dividing by TRUE leaves the value unchanged.

This technique is also easily modified to calculate a MINIF function by simply changing one number.

In the example on the right we have a list of regions in column A and their respective scores in column B.

	A	B	C	D	E
1	Region	Score	Region	High Score	
2	North	75%	North		75%
3	South	85%	South		85%
4	East	65%	East		95%
5	West	55%	West		55%
6	North	70%			
7	South	80%	Region	Low Score	
8	East	90%	North		60%
9	West	50%	South		80%
10	North	60%	East		65%
11	South	85%	West		45%
12	East	95%			
13	West	45%			


We can create our conditional MAX function to calculate the maximum scores for each region.

The formula in cell E2 is

=AGGREGATE(14,6,(\$B\$2:\$B\$13)/(\$A\$2:\$A\$13=D2),1)

This formula has been copied down for the other regions. The 14 represents the LARGE function and the 6 chooses to ignore errors. The 1 on the end specifies the largest value for the LARGE function.

The image below helps explain how the function works.



Each cell in the range (\$B\$2:\$B\$13) is divided by the result of the comparison of the corresponding row in column A as it is compared to cell D2 ("North"). The comparison returns TRUE or FALSE, hence the value from column B is divided by one or zero leaving valid scores and error messages.

Remember that the AGGREGATE function can ignore errors, hence only valid scores are left to be analysed by the LARGE function.

To calculate part of a formula select the part to analyse and press the F9 function key. This will display the results of that part of the formula.

To modify the function to identify the lowest values for a region, a MINIF function, you only need to change the 14 to 15. 15 is for the SMALL function - the opposite of the LARGE function. Cell E8 as an example of the MINIF formula.

=AGGREGATE(15,6,(\$B\$2:\$B\$13)/(\$A\$2:\$A\$13=D8),1)

This formula has been copied down for the other regions.

Pre-Excel 2007 solutions

The Dec_Array sheet has the array version of the formula. Array formulas have to be entered by holding Ctrl + Shift + Enter (CSE). This places braces { } around the formula. That is the visual clue a formula is an array formula. These formulas will work in all versions of Excel.

The formula in cell E2 is

{=MAX(IF(\$A\$2:\$A\$13=D2,\$B\$2:\$B\$13,""))}

The formula in cell E8 is

{=MIN(IF(\$A\$2:\$A\$13=D8,\$B\$2:\$B\$13,""))}

Most users are unaware of array formulas and the need to use CSE when entering for editing the formula. I tend to avoid using array formulas for that reason.

July 2013 - Pareto Chart


Pareto analysis involves the 80/20 rule which states (in general) that 80% of the outcomes are driven by 20% of the inputs.

In practical terms this may equate to

- 80% of your revenue coming from 20% of your customers
- 80% of your revenue coming from 20% of your products

The 80/20 rule has proven remarkably robust and has been shown to apply to many different fields and situations.

A Pareto chart is useful in helping visualise the impact of the 80/20 rule. A Pareto chart uses a left and a right axis. The left axis measures the value being analysed and the right axis is used to monitor the cumulative %. The example from the article is shown below.



From this chart you can identify that the top two reasons account for 60% of the defects. And the top four reasons account for 82% of the defects. Hence if you concentrate your efforts on reducing the top two reasons you will account for over half of all of the defects.

There are a couple of things you need to do to set the data up for a Pareto chart.

First you need to sort your items in descending order of the measure being reviewed.

Second you need to create a cumulative percentage of the measures.


The formula in cell C2 is

`=SUM(B2:B2)/SUM(B2:B10)`



This formula has been copied down the rest of the column. Note the use of the \$ signs to create the cumulative percentage.

To create the Pareto Chart

1. Select the range A1:C10.
2. Click the Insert tab and click the Column icon.
3. Select the top left icon for a Clustered Column chart.
4. In the Layout Ribbon tab click the drop down on the far left and select Series "Cumulative%"



5. Then click the Format Selection icon.
6. Select the Secondary Axis option and click Close.
7. Click the Design Ribbon tab and click the Change Chart Type icon on the far left and select the first Line chart and click OK.
8. That is the basic layout done - the rest is mainly formatting.
9. Change the left and right axis fonts to match the colours on the charts.
10. Add a Chart Title via the Layout Ribbon tab and link it to the cell A1 by selecting the Chart Title and pressing the = sign in the Formula Bar and clicking cell A1 and pressing Enter.



11. Move the legend to the top and resize it.
12. Expand the chart area as large as possible and drop the font size of bottom axis to fit.

September 2013 - NETWORKDAYS Function

Excel can work well with dates and working days.

The NETWORKDAYS function counts the number of workdays between two dates. It assumes a Saturday/Sunday weekend and also allows you to provide a list of holidays to be taken into account in the calculation of the number of workdays. It is an inclusive calculation including both the start and end dates in the calculation.

Excel 2010 extended the flexibility of the function with the new NETWORKDAYS.INTL function which has a flexible weekend that is determined by a number used within the function. This allows you to exclude a single day from your calculations. For example you might want to exclude Sundays only from your calculations.

The NETWORKDAYS function is related to the WORKDAY function. The WORKDAY function will calculate a date by adding a number of workdays to a start date taking into account weekends and holidays. Again in Excel 2010 a new function was added WORKDAY.INTL that allows you to define the weekend to include in the calculations.

These two functions work together enabling you to either figure out how many workdays between two dates or to figure out a completion date by adding a number of workdays to start that.

Syntax

NETWORKDAYS(Start_Date,End_Date,Holiday_Range)

Start_Date, End_Date - cells that contain dates

Holiday_Range (optional) - a range that contains a list of dates to be excluded from the calculation. If omitted it calculates weekdays.

NETWORKDAYS.INTL(Start_Date,End_Date,Weekend,Holiday_Range)

Weekend (optional) - a number that defines the days to be treated as the weekend. See list below for the numbers. If omitted it works just like the NETWORKDAYS function and uses Sat + Sun as the weekend.

fx =NETWORKDAYS.INTL(A2,C2,11,\$I\$2:\$I\$11)				
NETWORKDAYS.INTL(start_date, end_date, [weekend], [holidays])				
	End	End Date	1 - Saturday, Sunday	
		Day of the week	2 - Sunday, Monday	
			3 - Monday, Tuesday	
			4 - Tuesday, Wednesday	
			5 - Wednesday, Thursday	
			6 - Thursday, Friday	
			7 - Friday, Saturday	
			11 - Sunday only	
			12 - Monday only	
			13 - Tuesday only	
			14 - Wednesday only	
			15 - Thursday only	
	31/07/2013	Wednesday	21	20
	31/08/2013	Saturday		
	30/09/2013	Monday		
	31/10/2013	Thursday		
	30/11/2013	Saturday		
	31/12/2013	Tuesday		
	31/01/2014	Friday		
	28/02/2014	Friday		
	31/03/2014	Monday	25	

Examples

	A	B	C	D	E	F	G	H	I
1	Start Date	Start Date Day of the week	End	End Date Day of the week	Week Days	Work Days	No Sundays		Public Holidays (WA)
2	1/07/2013	Monday	31/07/2013	Wednesday	23	23	27		30/09/2013
3	1/08/2013	Thursday	31/08/2013	Saturday	22	22	27		25/12/2013
4	1/09/2013	Sunday	30/09/2013	Monday	21	20	24		26/12/2013
5	1/10/2013	Tuesday	31/10/2013	Thursday	23	23	27		1/01/2014
6	1/11/2013	Friday	30/11/2013	Saturday	21	21	26		27/01/2014
7	1/12/2013	Sunday	31/12/2013	Tuesday	22	20	24		3/03/2014
8	1/01/2014	Wednesday	31/01/2014	Friday	23	21	25		18/04/2014
9	1/02/2014	Saturday	28/02/2014	Friday	20	20	24		21/04/2014
10	1/03/2014	Saturday	31/03/2014	Monday	21	20	25		25/04/2014
11	1/04/2014	Tuesday	30/04/2014	Wednesday	22	19	23		2/06/2014
12	1/05/2014	Thursday	31/05/2014	Saturday	22	22	27		
13	1/06/2014	Sunday	30/06/2014	Monday	21	20	24		
14				Total	261	251	303		
16	1/07/2013	Monday	30/06/2014	Monday	261	251	303		

The formula in cell E2 is

=NETWORKDAYS(A2,C2)

This calculates weekdays because it omits the holiday range.

The formula in cell F2 is

=NETWORKDAYS(A2,C2,\$I\$2:\$I\$11)

This formula uses the holiday range and so omits those dates from the calculations.

The formula in cell G2 is

=NETWORKDAYS.INTL(A2,C2,11,\$I\$2:\$I\$11)

This formula uses the new function and 11 for the weekend option which uses Sunday as the weekend.

All these formulas have been copied down their columns.

Syntax

WORKDAY(Start_Date,Days,Holiday_Range)

Days - the number of days to add to the Start_Date.

WORKDAY.INTL(Start_Date,Days,Weekend,Holiday_Range)

The Start_Date is not included in the calculation, so if you add 1 it will give you the next working day.

If you want to include the start date then simply subtract 1 from the result.

October 2013 - Deleting Zero Rows Solutions

As with many things in Excel there are a few ways to delete zeros rows. Rather than delete the rows you may only want to hide them.

The type of solution you use will depend on the layout of your spreadsheet. If the data is in a table layout then you can use Excel's built-in filter feature. If the layout is less structured then you may need to use a macro. Depending on the situation you may want a non-macro solution and you can adapt the filter technique to work with unstructured sheets.

The first solution to this problem involves using Excel's filter feature. When you filter a data table Excel hides the rows that don't meet the criteria used in the filter. When dealing with a filtered list you don't impact the hidden rows. So if you delete the entries in a filtered list you are only deleting the visible rows.

Filter method

To apply a filter to a table select a cell in the table and press Ctrl + Shift + L, the same shortcut also turns off the filter.

You can use the filter feature to only show the zero rows. Then select the rows and delete them using Ctrl + - (minus sign). Be very careful if you see the Delete dialog as you may not be deleting what you want.

Macro Solution to Delete

WARNING: MACROS CANNOT BE UNDONE!!!

```
Sub Delete_Zero_Rows()
'this macro deletes rows if the values in columns D and E
'in those rows are both zero or blank
Dim r As Long
Dim LastRow As Long

Application.ScreenUpdating = False
Application.Calculation = xlCalculationManual

LastRow = ActiveSheet.UsedRange.Rows.Count

For r = LastRow To 2 Step -1
    If Cells(r, "D").Value = 0 And Cells(r, "E").Value = 0 Then
        Rows(r).Delete
    End If
Next r

Application.ScreenUpdating = True
Application.Calculation = xlCalculationAutomatic

End Sub
```

This is not a recorded macro - it has been written from scratch.

The macro solution uses the *r* variable to go through each used row and determine whether or not to delete it. The *r* variable is used to hold the row number.

The For - Next loop starts at the LastRow value and works down to 2.

The LastRow variable captures the final used row in the sheet. When deleting rows you need to start at the bottom and work your way up the sheet. If you start at the top and work your way down then you will miss rows.

The two Application commands deal with the screen updating (stopping the screen flicker) and turning calculation off to speed up the process and then turn it back on at the end.

Hiding rows via a macro

If you wanted to hide the rows instead of deleting them then you could amend the macro as follows.

The changes are marked.

```
Sub Hide_Zero_Rows()  
'this macro deletes rows if the values in columns D and E  
'in those rows are both zero or blank  
Dim r As Long  
Dim LastRow As Long  
  
Application.ScreenUpdating = False  
Application.Calculation = xlCalculationManual  
  
LastRow = ActiveSheet.UsedRange.Rows.Count  
  
For r = LastRow To 2 Step -1  
    If Cells(r, "D").Value = 0 And Cells(r, "E").Value = 0 Then  
        Rows(r).Hidden = True  
    Else  
        Rows(r).Hidden = False  
    End If  
Next r  
  
Application.ScreenUpdating = True  
Application.Calculation = xlCalculationAutomatic  
  
End Sub
```

Both macros are in the companion file.

Appendix - Shortcuts

Each of my Excel Yourself articles in 2013 contained a tip at the end of the article - here they all are.

Alt + ;

This shortcut selects visible cells only. This is useful if you need to capture just the visible cells when you have hidden rows or columns or when you have used Grouping to hide rows and columns. Be warned that formulas do not paste. Only their values are pasted. Note if you use filtering to hide rows then when you copy a filtered list you will only paste the visible cells anyway.

Ctrl + Enter

This shortcut allows you to enter the same value or formula in a range. First select a range, then type your entry and then hold Ctrl and press Enter and that will populate the whole range with your entry.

+ Sign

You can start all your formulas with +. I do this all the time as the + has a large key on the far right of the keyboard which is easy to select and faster than pressing the = key which is tucked away in the middle of the keyboard. When you press Enter or the arrow keys Excel will automatically insert the = in front of the +.

Ctrl + *

This shortcut selects what is called the current region. In Excel the current region is the region that contains entries that are bound by blank rows and blank columns. The shortcut is excellent for selecting tables. It will select a table no matter how long that table is providing you are in a cell within the table. It is similar to using Ctrl + a, except if you are not in a table, Ctrl + a will select everything whereas Ctrl + * doesn't change the current selection.

Shift + Ctrl + ➡

This one allows you to select all the cells to the right until a blank cell is encountered. Holding Ctrl and the Shift key down whilst using the arrow keys is a very quick way to select larger ranges. Again it works very well with table layouts.

Undo - the undo command in Excel was extended in Excel 2007 to enable you to Undo entries after you have saved.

Alt + v v

Pressed in sequence, not held down - this opens the Data Validation dialog in Excel.

Alt + w f f

Pressed in sequence this will turn on or turn off the Freeze Panes settings on the current sheet. Freeze Panes settings allow you to fix rows and columns so they don't change as you scroll around your spreadsheet.

Alt h w

Pressed in sequence, is the shortcut for the Wrap Text format to the current selection. Wrap Text will insert lines within a cell to enable you to display more text without having to widen the column. This is useful for the heading rows in data tables.

Shift to Insert

This shortcut involves holding the Shift key down whilst you move a column or a row. If you try and move a column using the arrow shaped cursor you will in fact over write your destination column. If you hold the Shift key down whilst you drag the column you will insert the column between the existing columns. **Warning:** this may cause #REF! Errors depending on your formulas.

Shift + Alt + Right Arrow**Shift + Alt + Left Arrow**

You can easily apply Grouping (hiding rows/columns) by using the Shift + Alt + Right Arrow. You can remove Grouping by using the Shift + Alt + Left Arrow. Grouping is the preferred way to hide rows or columns because it adds icons around the outside of the grid which show that rows and columns have been hidden. It is also very quick to hide and unhide using those icons.